
はじめに

 2群間の比較においてWilcoxon順位和検定を適用した場合、本検定に

対応する推定方法としてはHodges-Lehmann 推定量が有名であるが、

解析を行う状況によっては本推定方法が適切でない場合がある

 本発表では、Hodges-Lehmann 推定量の復習と推定が悪くなる例を

紹介した後、Wilcoxon-Mann-Whitneyオッズ（WMWodds）による

推定方法を紹介する。次に、いくつかのシミュレーションを行い、

Hodges–Lehmann による推定との比較を行う

Wilcoxon 順位和検定

 投与群が 2群ある臨床試験を考える

 群 1（被験薬群）の応答変数：𝑌1𝑖 (i = 1, … , 𝑛1)

 群2（対照薬群）の応答変数：𝑌2𝑖 (i = 1, … , 𝑛2)

 各投与群の母集団分布の確率密度関数をそれぞれ f x 、g(x)と表す

 Wilcoxon順位和検定（又は、Mann–Whitneyの U検定）は、

2つの分布の形状は同じだが位置がある定数 δだけずれている、

すなわち g(x) = f(x + δ)を仮定し、以下の仮説について検定を行う

 
帰無仮説 H0: δ = 0

対立仮説 H1: δ ≠ 0

Hodges-Lehmann 型の推定方法

 Wilcoxon 検定に対する推定方法としては、Hodges-Lehmann型の

推定方法が一般的に用いられる

 𝑛1𝑛2個の 𝑑𝑖𝑗 = 𝑌1𝑖 − 𝑌2𝑗を生成する

 𝑑𝑖𝑗を小さい順に並べたものを 𝑑(1) ≤ 𝑑 2 ≤ ⋯ ≤ 𝑑 𝑛1𝑛2
とし、

𝑐𝛼
∗ =

𝑛1𝑛2

2
− 𝑧1−𝛼/2

𝑛1𝑛2(𝑛1+𝑛2+1)

12
とする

 𝛿の点推定値：生成した 𝑑𝑖𝑗の中央値

 𝛿の両側 95%信頼区間：

 両側 95%信頼区間の下限 = 𝑑(𝑐𝛼)

 両側 95%信頼区間の上限 = 𝑑(𝑛1𝑛2+1−𝑐𝛼)

※ 𝑐𝛼：𝑐𝛼
∗以下の値のうち最も大きい整数値

例 2.1

 各投与群の応答変数の分布と要約統計量が下記である場合を考える

 すなわち、標本数が大きく、応答変数の実現値は整数であり、分布の

中央値付近の密度が両群とも高く、分布の裾が一方の分布だけ広がる

場合を考える

平均値= -0.1
中央値= 0

平均値= -0.4
中央値= 0



例 2.1

 npar1wayプロシジャ+ hlオプションにてWilcoxon順位和検定及び

Hodges-Lehmann型による δの推定が実行できる（以下は結果の抜粋）

 Wilcoxon順位和検定の結果は p=0.0471となり有意差がみられるが、

δの点推定値及び両側 95%信頼区間の推定結果は全て 0となり、

解釈が難しい結果となる

 特に、両側 95%信頼区間が［0, 0］となるのは問題である。

Wilcoxon の順位和検定( 2 標本)

正規近似
Z 1.9855
両側 Pr > |Z| 0.0471

Hodges-Lehmann 推定値
Location Shift 0.0000

95% 信頼限界 区間の中間点 漸近標準誤差
0.0000 0.0000 0.0000 0.0000

Wilcoxon-Mann-Whitneyオッズ（WMWodds）

 群 1のデータ 𝑦11, … , 𝑦1𝑛1
と群 2のデータ 𝑦21, … , 𝑦2𝑛2

を小さい順に並べ

「 𝑦1𝑖未満の群 2のデータ数」を 𝑢𝑖としたとき、

 𝑈 =  𝑖=1
𝑛1 𝑢𝑖がMann-Whitneyの U統計量となる

 同じ値のデータ（タイデータ）がある場合は「 𝑦1𝑖未満の群 2のデータ数」に

「 𝑦1𝑖と同じ値である群 2のデータ数の 1/2」を加えて 𝑢𝑖とする

 #(条件式)を「条件式を満たす要素の数」とし、以下のように略記する

𝑈 = #(𝑌1 > 𝑌2) +
1

2
× #(𝑌1 = 𝑌2)

 Mann-Whitneyの U統計量とWilcoxon順位和検定の検定統計量Wとの

間には以下の関係が成り立ち、本質的に 2つの統計量は同等である

𝑊 = 𝑈 +
𝑛1(𝑛1 + 1)

2

Wilcoxon-Mann-Whitneyオッズ（WMWodds）

 Wilcoxon 検定と、以下の仮説に関する検定とは同等である

 
帰無仮説𝐻0: 𝜋 = 0.5

対立仮説𝐻1: 𝜋 ≠ 0.5
, 𝜋 = 𝑃 𝑌1 > 𝑌2 +

1

2
𝑃 𝑌1 = 𝑌2 =

𝑈

𝑛1𝑛2

 O'Brien（2006）は、上記の πに関するオッズを定義し、これを

WMWoddsとした

WMWodds =
𝜋

1 − 𝜋

 例えば、「WMWodds=2」は以下の様に解釈出来る

「群 1からランダムに選んだ標本は、群 2からランダムに選んだ標本よりも

確率的に大きく、その度合いはオッズの尺度で 2である。」

Wilcoxon-Mann-Whitneyオッズ（WMWodds）

 O'Brien（2006）ではWMWoddsの点推定値とその両側 95% 信頼区間の

計算方法と SAS マクロが提案されているが、WMWoddsの計算のための

SASマクロのリンクが切れており、現在は入手できない

 そこで本発表では、O'Brien（2006）で紹介されている方法※よりも

簡便に計算を行う目的で、Hanley（1982）、Mason（2002）及び

Acion（2006）等で紹介されている「Mann-Whitneyの U統計量と、

ROC曲線の曲線下面積（以下、AUROC）との関係」を用いて、

WMWoddsの点推定値とその両側 95%信頼区間の計算を行う

11 ※参考的に O'Brien（2006）で紹介されている
方法の概要を Backup Slides に含めた

WMWoddsと ROC曲線下面積（AUROC）との関係

 いま想定している臨床試験について、投与群（ 2群）を従属変数、

応答変数を独立変数とした ROC曲線を考えると、Mann-Whitneyの

U統計量と AUROCとの間には以下の関係が成り立つ※

AUROC =
𝑈

𝑛1𝑛2
(= 𝜋)

 よって、WMWodds =
𝜋

1−𝜋
と AUROC との間には以下の関係が成り立つ

WMWodds =
AUROC

1 − AUROC

 また、WMWoddsに関する両側 95%信頼区間は以下により計算できる

exp log
𝜋

1 − 𝜋
− 𝑧0.975

𝜎

𝜋(1 − 𝜋)
, exp log

𝜋

1 − 𝜋
+ 𝑧0.975

𝜎

𝜋(1 − 𝜋)

 筆者が作成した SASマクロ「%WMWodds」にて計算できる

13 ※証明の概要を Backup Slides に含めた



SASマクロ「%WMWodds」
%macro WMWodds(_DATASET, _GROUP, _Y) ;
ods listing close ;

proc npar1way data=&_DATASET. correct=no ;
class &_GROUP.  ;
var &_Y. ;
ods output WilcoxonScores=WS(keep=Class SumOfScores) ;

run ;

proc transpose data=WS out=FLAG prefix=WSCORE_ ;
var SumOfScores ;
id  CLASS ;

run ;

data _NULL_ ;
set FLAG ;
if (WSCORE_1 >= WSCORE_2) then call symput("_FLAG",1);
else                       call symput("_FLAG",2);

run ;

proc logistic data=&_DATASET. ;
class &_GROUP. ;
model &_GROUP.=&_Y. ;
roc ;
ods output ROCAssociation=AUROC ;

run ;

ods listing ;   

data WMWODDS ;
set AUROC ;
if (&_FLAG=1) then do ;
WMWOdds=Area/(1-Area) ;
SE     =StdErr/(1-Area)**2 ;
LnWMW =log(Area/(1-Area)) ;
LnSE =StdErr/(Area*(1-Area)) ;

end ;
else do ;
WMWOdds=(1-Area)/Area ;
SE     =StdErr/Area**2 ;
LnWMW =log((1-Area)/Area) ;
LnSE =StdErr/(Area*(1-Area)) ;

end ;

z          =quantile("NORMAL",0.975) ;
LowerCI =WMWOdds - z*SE ;
UpperCI =WMWOdds + z*SE ;
LowerCI_exp=exp(LnWMW - z*LnSE) ;
UpperCI_exp=exp(LnWMW + z*LnSE) ;

if _n_=1;
keep Area StdErr WMWOdds SE LowerCI UpperCI

LowerCI_exp UpperCI_exp ;
run ;

title "&_DATASET." ;
proc print noobs ;
run ;

%mend ;

計算例

 タイがない場合とある場合の 2つのデータに対して計算例を紹介する

15

投与群 Y 順位 ui

2 11 1

1 12 2 1

2 13 3

1 14 4 2

2 15 5

1 16 6 3

1 17 7 3

投与群 Y 順位 ui

2 11 1

1 12 2 1

2 13 3.5

1 13 3.5 1.5

2 15 5

1 16 6 3

1 17 7 3

表5.1 タイのないデータ 表5.2 タイのあるデータ

𝑈5.1 = 1 + 2 + 3 + 3 = 9 , 𝜋5.1 =
𝑈5.1

3 × 4
= 0.7500 , WMWodds5.1 =

0.7500

1 − 0.7500
= 3.0000

𝑈5.2 = 1 + 1.5 + 3 + 3 = 8.5 , 𝜋5.2 =
𝑈5.2

3 × 4
= 0.7083 , WMWodds5.2 =

0.7083

1 − 0.7083
= 2.4286

※上記の 𝜋5.1と 𝜋5.2は、AUROCと一致する



計算例

 U統計量、AUROCとその標準誤差を算出するプログラムを紹介する

data SAMPLEDATA2 ;
input TREAT Y ;
cards ;

2 11
1 12
2 13
1 14
2 15
1 16
1 17

;
run ;

proc npar1way data=SAMPLEDATA2 correct=no ;
class TREAT  ;
var Y ;

run ;

proc logistic data=SAMPLEDATA2 ;
class TREAT ;
model TREAT=Y ;
roc ;

run ;

data SAMPLEDATA3 ;
input TREAT Y ;
cards ;

2 11
1 12
2 13
1 13
2 15
1 16
1 17

;
run ;

proc npar1way data=SAMPLEDATA3 correct=no ;
class TREAT  ;
var Y ;

run ;

proc logistic data=SAMPLEDATA3 ;
class TREAT ;
model TREAT=Y ;
roc ;

run ;

プログラム 5.1（タイのないデータ） プログラム 5.2（タイのあるデータ）



計算例

 表 5.1と表 5.2のデータに対して SASマクロ「%WMWodds」を適用し、

それぞれのWMWoddsを計算する

%WMWodds(SAMPLEDATA2, TREAT, Y) ; *--- 表5.1 ;

%WMWodds(SAMPLEDATA3, TREAT, Y) ; *--- 表5.2 ;

 Area：AUROCの点推定値

 StdErr：AUROCの標準誤差

 WMWOdds：WMWoddsの点推定値

 SE：WMWoddsの標準誤差

 LowerCI_exp：WMWoddsの

両側95%信頼区間の下限

 UpperCI_exp：WMWoddsの

両側95%信頼区間の上限

手法の比較①：Wilcoxon 順位和検定の結果との対応

 各群の応答変数 𝑌1𝑖及び 𝑌2𝑖に対して同じ確率分布を仮定し、δ = 1だけ

ずらしたシミュレーションデータを用いて、Hodges-Lehmann型による

δの両側 95% 信頼区間とWMWoddsの両側 95%信頼区間について、

Wilcoxon 順位和検定の結果とどれだけ対応が取れているかを調査する

 各群の例数：250例

 確率分布：正規分布、指数分布、ポアソン分布及び負の二項分布

 場面：それぞれの確率分布について、Wilcoxon順位和検定の p値が

0.05をわずかに下回る（有意差あり）場合と

0.05 をわずかに上回る（有意差あり）場合の 2パターンを用意

手法の比較①：Wilcoxon 順位和検定の結果との対応

手法の比較①：Wilcoxon 順位和検定の結果との対応

21

確率分布
（1群250例）

Wilcoxon
順位和検定

Hodges-Lehmann 型 WMWodds

Lower Upper Lower Upper

正規分布
 𝛿 = 0.9

𝑊𝑀𝑊𝑜𝑑𝑑𝑠 = 1.2

p=0.0480 0.0109 1.8471 1.0011 1.5055

p=0.0494 0.0028 1.8541 0.9999 1.5037

p=0.0502 -0.0012 1.8575 0.9991 1.5024

p=0.0512 -0.0053 1.8610 0.9982 1.5011

指数分布
 𝛿 = 0.9

𝑊𝑀𝑊𝑜𝑑𝑑𝑠 = 1.2

p=0.0494 0.0010 1.8483 1.0001 1.5034

p=0.0497 0.0005 1.8487 0.9997 1.5029

p=0.0501 -0.0009 1.8499 0.9994 1.5025

p=0.0515 -0.0062 1.8544 0.9981 1.5005

ポアソン分布
 𝛿 = 1.0

𝑊𝑀𝑊𝑜𝑑𝑑𝑠 = 1.2

p=0.0490 0.0000 2.0000 1.0006 1.5021

p=0.0495 0.0000 2.0000 1.0001 1.5013

p=0.0502 0.0000 2.0000 0.9995 1.5005

p=0.0539 0.0000 2.0000 0.9963 1.4956

負の二項分布
 𝛿 = 3.0

𝑊𝑀𝑊𝑜𝑑𝑑𝑠 = 1.2

p=0.0481 0.0000 6.0000 1.0014 1.5047

p=0.0494 0.0000 6.0000 1.0002 1.5029

p=0.0508 0.0000 6.0000 0.9990 1.5010

p=0.0515 0.0000 6.0000 0.9984 1.5001

赤字下線部：有意差あり



手法の比較①：Wilcoxon 順位和検定の結果との対応

 データが連続分布に従っている状況では（タイが生じにくい状況では）、

「 2つの分布の形状は同じだが位置がある定数 δだけずれている」という

仮定が成り立っていれば、Hodges-Lehmann型による δの両側 95%信頼区間

は良好な結果となることが分かり、WMWoddsの両側 95%信頼区間では、

p値が 0.05をわずかに下回る状況では Hodges-Lehmann型による推定よりも

わずかに劣ることが示唆された。

（ただし、この問題は実用上はほとんど気にならないと考えられる）

 データが離散分布に従っている状況では、タイが生じやすい状況である

ため、 𝑑𝑖𝑗 = 𝑌1𝑖 − 𝑌2𝑗に基づいて δの推定を行う Hodges-Lehmann型の方法

では、𝑑𝑖𝑗として同じ値ばかりが生成されるため望ましい結果が得られない

ことが伺える。

⇒ さらに考察するため、先ほど用いた正規分布及び指数分布に従うデータ

の小数点以下を切り捨て（整数化し）、このデータに対して同様の

シミュレーションを行った。

手法の比較①：Wilcoxon 順位和検定の結果との対応

手法の比較①：Wilcoxon 順位和検定の結果との対応

 Hodges-Lehmann 型による δの両側 95% 信頼区間の下限は全て 0。

 WMWoddsの両側 95%信頼区間はWilcoxon順位和検定の結果と対応が

良く取れていることから、データの分布が連続分布でも離散分布でも、

Wilcoxon順位和検定の結果と対応が良く取れていることが分かる。

24

確率分布
（1群250例）

Wilcoxon
順位和検定

Hodges-Lehmann 型 WMWodds

Lower Upper Lower Upper

正規分布
 𝛿 = 1.0

𝑊𝑀𝑊𝑜𝑑𝑑𝑠 = 1.2

p=0.0441 0.0000 2.0000 1.0049 1.5110

p=0.0558 0.0000 2.0000 0.9945 1.4941

p=0.0535 0.0000 2.0000 0.9964 1.4971

p=0.0501 0.0000 2.0000 0.9992 1.5015

指数分布
 𝛿 = 1.0

𝑊𝑀𝑊𝑜𝑑𝑑𝑠 = 1.2

p=0.0523 0.0000 2.0000 0.9975 1.4982

p=0.0509 0.0000 2.0000 0.9987 1.5000

p=0.0492 0.0000 2.0000 1.0002 1.5023

p=0.0564 0.0000 2.0000 0.9941 1.4930

赤字下線部：有意差あり

手法の比較②：両側 95% 信頼区間の被覆確率

 前項と同様の状況ではあるが、本項では各群の応答変数 𝑌1𝑖及び 𝑌2𝑖に

対して全く同じ確率分布を仮定して（δ = 0として）シミュレーション

データを生成し、Hodges-Lehmann型による δの両側 95%信頼区間と

WMWoddsの両側 95%信頼区間について被覆確率に関する調査を行う

 Hodges-Lehmann型：両側 95%信頼区間が 0を含んでいる確率

 WMWodds：両側 95%信頼区間が 1を含んでいる確率

 各群の例数：10、20、50、100及び 200例

 確率分布：正規分布、指数分布、ポアソン分布及び負の二項分布

 シミュレーション回数：5000 回

26 参考までに、「Wilcoxon順位和検定の結果では有意差なしだが、各手法の
両側 95%信頼区間では有意差ありとなる確率」は、いずれも 0.0%であった。



正規分布及び指数分布に関する被覆確率
〔黒：Hodges-Lehmann 型、赤：WMWodds〕

 両手法とも 95%を下回る場合が散見されたが、WMWoddsの両側 95%
信頼区間の方が頻度は小さかった。

 また、WMWoddsの被覆確率の範囲は 94.7%～97.6%であり、第 1種の
過誤確率は概ね 5%以内に抑えられており、かつ過度に保守的になって
いないことが示唆された。

ポアソン分布及び負の二項分布に関する被覆確率
〔黒：Hodges-Lehmann 型、赤：WMWodds〕

負の二項分布に関する被覆確率
〔黒：Hodges-Lehmann 型、赤：WMWodds〕

 Hodges-Lehmann 型の被覆確率は過度に大きくなった。これは前項の考

察より、離散分布に対するHodges-Lehmann型による δの両側 95%信頼

区間の下限は 0になりやすいことが原因と思われる。

 WMWoddsの被覆確率は 95%を下回る場合が散見されたが、被覆確率の

範囲は 94.3%～97.2%であり、第 1種の過誤確率は概ね5%以内に抑え

られており、かつ過度に保守的になっていないことが示唆された。

メニュー

1. Wilcoxon順位和検定と Hodges-Lehmann型の推定方法

2. Wilcoxon-Mann-Whitneyオッズ（WMWodds）

3. WMWoddsと ROC曲線下面積（AUROC）との関係

4. 手法の比較①：Wilcoxon順位和検定の結果との対応

5. 手法の比較②：両側 95% 信頼区間の被覆確率

6. まとめ

まとめ

「Wilcoxon順位和検定結果との対応」と「両側95%信頼区間の被覆確率」

の観点から、連続分布と離散分布の場合においてHodges-Lehmann型に

よる δの両側 95%信頼区間とWMWoddsとその両側95%信頼区間の比較

を行った。

Hodges-Lehmann型による両側 95%信頼区間

データが連続分布に従っている場合は望ましいが、データが離散分布に従って

いる場合やタイが生じやすい状況では性能が悪くなることが示唆された。

WMWoddsの両側95%信頼区間の推定結果

データの分布が連続分布であっても離散分布であっても望ましい結果となる

ことが分かった。また、WMWoddsの両側 95%信頼区間の被覆確率は概ね

95%を上回っており、第 1種の過誤確率の観点からもWMWoddsの両側 95%

信頼区間は望ましい性質を持つことが分かった。

Mann-Whitneyの U統計量と AUROCとの関係

タイがない場合

 AUROC = 1 −
1

𝑛1𝑛2
 𝑖=1

𝑛1 𝑓𝑖 (𝑓𝑖 ∶ 𝑦1𝑖より大きい群 2のデータ数)

 Mann−Whitneyの U =  𝑖=1
𝑛1 𝑣𝑖 (𝑣𝑖 ∶ 𝑦1𝑖より小さい群 2のデータ数)

 𝑛1𝑛2 =  𝑖=1
𝑛1 𝑓𝑖 +  𝑖=1

𝑛1 𝑣𝑖 となるので、よって AUROC =
𝑈

𝑛1𝑛2

タイがある場合

 AUROC = 1 −
1

𝑛1𝑛2
 𝑖=1

𝑛1 𝑓𝑖 −
1

2𝑛1𝑛2
 𝑖=1

𝑛1  𝑓𝑖 − 𝑓𝑖 ( 𝑓𝑖 ∶ 𝑦1𝑖以上の群 2のデータ数)

 Mann−Whitneyの U =  𝑖=1
𝑛1 𝑣𝑖 +

1

2
 𝑖=1

𝑛1  𝑓𝑖 − 𝑓𝑖

 𝑛1𝑛2 =  𝑖=1
𝑛1 𝑓𝑖 +  𝑖=1

𝑛1 𝑣𝑖 +  𝑖=1
𝑛1  𝑓𝑖 − 𝑓𝑖 となるので、よって AUROC =

𝑈

𝑛1𝑛2

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

 AUROC = 1 −
1

𝑛1𝑛2
 𝑖=1

𝑛1 𝑓𝑖 −
1

2𝑛1𝑛2
 𝑖=1

𝑛1  𝑓𝑖 − 𝑓𝑖

 𝑈5.2 = 1 + 1.5 + 3 + 3 = 8.5 , 𝜋5.2 =
𝑈5.2

3×4
= 0.708

= 1 −
1

1

2×3×4
0 + 0 + 1 + 0 =

17

24
= 0.708

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

Mann-Whitneyの U統計量と AUROCとの関係

generalized odds ratio（genOR）

 O'Brien（2006）では、WMWoddsの標準誤差は Agresti（1980）にて

紹介されている generalized odds ratio（genOR）を基に算出している

 genORの定義は以下の通り

 genOR =
Prob Y1> Y2

Prob Y1<Y2
=

1−γ

1+γ

 γは Goodman-Kruskall gamma 統計量であり、freqプロシジャで推定値及び

標準誤差の算出が可能

 タイデータが存在しなければ、 genOR = WMWoddsとなるため、

γの標準誤差とデルタ法を用いて genORの標準誤差が算出できる

 𝑉 genOR =
2

1− γ 2 𝑉  γ

54



generalized odds ratio（genOR）

 タイデータが存在する場合、以下の手順で変換を行う

① 値に対する順位付けを行い、頻度集計を行う。

② タイデータが存在する順位について、片方の群の順位を 2つの順位に

分割する。このとき、分割された順位は元の順位に±0.1をそれぞれ

加えたものとなる。

③ 分割された順位の頻度は元の頻度の 1/2にする。

 上記の変換を行った上で、 genORの標準誤差を算出する

 次頁以降では、genORを基にWMWoddsの両側 95% 信頼区間を算出

した場合の（すなわち、O'Brien（2006）で実施していたと予想される

計算方法での）シミュレーション結果を紹介する

generalized odds ratio（genOR）

シミュレーション回数は 5000回、確率は小数第 2位を四捨五入して表示し、95%未満のものに下線を引いた。

確率分布 1群あたりの例数 分布のパラメータ Hodges-Lehmann型
WMWodds

本発表の方法 O'Brien

正規分布

10例
分散＝1 95.4% 97.0% 95.6%
分散＝5 96.2% 97.6% 96.7%
分散＝9 96.3% 97.4% 96.5%

20例
分散＝1 94.9% 95.6% 95.1%
分散＝5 95.4% 96.2% 95.5%
分散＝9 95.4% 96.2% 95.6%

50例
分散＝1 94.9% 95.2% 95.0%
分散＝5 95.1% 95.6% 95.3%
分散＝9 95.9% 96.1% 95.8%

100例
分散＝1 95.1% 95.3% 95.2%
分散＝5 94.6% 94.8% 94.7%
分散＝9 95.0% 95.2% 95.1%

200例
分散＝1 95.1% 95.2% 95.1%
分散＝5 95.3% 95.4% 95.3%
分散＝9 94.7% 94.7% 94.7%

指数分布

10例
平均＝1 95.6% 97.0% 96.0%
平均＝5 95.8% 97.3% 96.1%
平均＝9 95.8% 97.4% 96.2%

20例
平均＝1 95.3% 95.9% 95.4%
平均＝5 95.2% 95.9% 95.3%
平均＝9 94.9% 95.5% 94.9%

50例
平均＝1 94.7% 95.1% 94.8%
平均＝5 95.6% 95.9% 95.7%
平均＝9 94.8% 95.0% 95.0%

100例
平均＝1 94.9% 95.1% 94.9%
平均＝5 95.2% 95.3% 95.2%
平均＝9 94.4% 94.7% 94.5%

200例
平均＝1 95.3% 95.3% 95.3%
平均＝5 94.8% 95.0% 94.9%
平均＝9 94.8% 94.8% 94.8%



generalized odds ratio（genOR）

シミュレーション回数は 5000回、確率は小数第 2位を四捨五入して表示し、95%未満のものに下線を引いた。

確率分布 1群あたりの例数 分布のパラメータ Hodges-Lehmann型
WMWodds

本発表の方法 O'Brien

ポアソン分布

10例

平均＝1 99.8% 96.1% 97.4%
平均＝5 98.9% 96.9% 96.5%
平均＝9 98.3% 97.2% 96.3%

20例

平均＝1 99.9% 95.2% 96.5%
平均＝5 99.4% 95.9% 95.8%
平均＝9 98.4% 95.2% 95.0%

50例

平均＝1 100.0% 95.1% 96.7%
平均＝5 99.8% 96.0% 96.1%
平均＝9 99.5% 95.3% 95.3%

100例

平均＝1 100.0% 94.9% 96.5%
平均＝5 99.9% 95.7% 95.9%
平均＝9 99.8% 94.8% 94.9%

200例

平均＝1 100.0% 95.6% 97.0%
平均＝5 100.0% 95.1% 95.4%
平均＝9 99.9% 95.0% 95.2%

負の二項分布
（成功回数：k ）

10例

p=0.25, k＝1 98.8% 97.1% 96.7%
p=0.25, k＝5 97.1% 97.0% 95.9%
p=0.25, k＝9 96.3% 96.7% 95.7%

20例

p=0.25, k＝1 99.2% 95.9% 95.9%
p=0.25, k＝5 97.0% 95.7% 95.1%
p=0.25, k＝9 96.6% 96.0% 95.6%

50例

p=0.25, k＝1 99.8% 94.9% 95.2%
p=0.25, k＝5 97.8% 95.2% 95.0%
p=0.25, k＝9 97.1% 95.2% 95.0%

100例

p=0.25, k＝1 100.0% 94.9% 95.3%
p=0.25, k＝5 98.5% 94.5% 94.5%
p=0.25, k＝9 97.9% 95.6% 95.5%

200例

p=0.25, k＝1 100.0% 95.4% 95.8%
p=0.25, k＝5 99.3% 95.8% 95.7%
p=0.25, k＝9 98.6% 95.2% 95.2%



generalized odds ratio（genOR）

シミュレーション回数は 5000回、確率は小数第 2位を四捨五入して表示し、95%未満のものに下線を引いた。

確率分布 1群あたりの例数 分布のパラメータ Hodges-Lehmann型
WMWodds

本発表の方法 O'Brien

負の二項分布
（成功回数：k ）

10例

p=0.50, k＝1 99.8% 96.7% 98.1%
p=0.50, k＝5 97.8% 96.8% 96.1%
p=0.50, k＝9 98.1% 97.1% 96.3%

20例

p=0.50, k＝1 100.0% 96.4% 98.0%
p=0.50, k＝5 98.7% 96.0% 95.8%
p=0.50, k＝9 98.2% 95.8% 95.6%

50例

p=0.50, k＝1 100.0% 95.4% 97.2%
p=0.50, k＝5 99.6% 95.7% 95.6%
p=0.50, k＝9 99.2% 95.8% 95.7%

100例

p=0.50, k＝1 100.0% 95.4% 97.4%
p=0.50, k＝5 99.8% 94.4% 94.6%
p=0.50, k＝9 99.3% 94.5% 94.5%

200例

p=0.50, k＝1 100.0% 95.0% 97.4%
p=0.50, k＝5 100.0% 95.0% 95.1%
p=0.50, k＝9 99.9% 95.1% 95.3%

10例

p=0.75, k＝1 100.0% 96.3% 99.7%
p=0.75, k＝5 99.4% 96.6% 96.8%
p=0.75, k＝9 98.9% 96.8% 96.6%

20例

p=0.75, k＝1 100.0% 94.4% 99.5%
p=0.75, k＝5 99.7% 95.9% 96.4%
p=0.75, k＝9 99.3% 95.8% 95.8%

50例

p=0.75, k＝1 100.0% 94.9% 99.5%
p=0.75, k＝5 100.0% 95.9% 96.5%
p=0.75, k＝9 99.9% 95.3% 95.6%

100例

p=0.75, k＝1 100.0% 94.8% 99.5%
p=0.75, k＝5 100.0% 94.3% 95.4%
p=0.75, k＝9 100.0% 95.0% 95.5%

200例

p=0.75, k＝1 100.0% 94.7% 99.6%
p=0.75, k＝5 100.0% 95.1% 96.0%
p=0.75, k＝9 100.0% 95.1% 95.5%


